



BOSCH

Invented for life

Aula 5 - Bluetooth

Docupedia Export

Author: Siqueira Joao (CtP/ETS)
Date: 02-Mar-2023 14:44

Table of Contents

1 Visão geral	4
2 Algumas aplicações do BLE	5
3 Comparativo	6
4 Código	7
5 Conectando com Smartphone	10



O **BLE** é um protocolo ligeiramente diferente do Bluetooth tradicional. Ele é mais adequado para aplicações **IoT** (Internet of Things) de **baixo consumo de energia**, já que a capacidade dos pacotes de dados é baixa.

Os dispositivos que trabalham com BLE podem ter duas funções diferentes em uma conexão: **Dispositivo Central ou Dispositivo Periférico** (Central Device or Peripheral Device). Geralmente, os dispositivos centrais são telefones celulares, tablets, computadores, etc. São dispositivos centrais que recebem dados. Já os dispositivos periféricos são sensores e dispositivos low power que se conectam ao dispositivo central. Podemos pensar também como uma estrutura cliente/servidor, onde um celular é o cliente e o sensor é o servidor que “serve” seus dados para o cliente.

GATT (Generic Attribute Profile) é a forma com que os dados são organizados para comunicação entre os dispositivos. GATT é composto por um ou mais serviços que, por sua vez, são compostos de características. Existem especificações padrão de GATT para os tipos de aplicação mais comuns encontradas no mercado. Várias dessas especificações podem ser encontradas no site oficial do Bluetooth. As características, por sua vez, são basicamente os valores em si.

Os serviços e as características são identificados por um **UUID** (universally unique identifier). Assim podemos customizar nossas próprias ou utilizar as já existentes. <https://www.bluetooth.com/specifications/gatt/services>

Como dito anteriormente, **podemos ter um ou mais serviços rodando no mesmo dispositivo** e, cada um deles, com suas próprias características.

Por exemplo, um ESP32 pode estar rodando um serviço de CLIMA, para enviar dados de temperatura e umidade, e um outro serviço de BATERIA, para enviar dados de medição de carga da bateria.

1 Visão geral



Podemos enviar e receber dados em ambos os lados.

Exemplo: ESP32 pode enviar dados de sensores.

Exemplo 2: Smartphone pode enviar comandos para o ESP32 ligar um relé.

2 Algumas aplicações do BLE

- Sensores de batimento cardíaco
 - Comunicação M2M (máquina-a-máquina)
 - Automatização de tarefas em uma residência
 - Controle remoto
 - Notificação de alerta de proximidade
 - dentre outros
-

3 Comparativo

	Bluetooth clássico	Bluetooth Low Energy
Preço	Alto	Muito baixo
Consumo de energia	Alto	Muito baixo
Aplicação	Transferências com grandes quantidades de dados	Trasferências de pequenas quantidades de dados



4 Código

Faremos um programa no qual o **ESP32** se tornará um **servidor BLE**. Quando um dispositivo estiver conectado a ele serão enviados em intervalos de tempo dados de um sensor de temperatura. O ESP32 também poderá receber dados para comandar o Liga/Desliga de um LED.

```

1  #include <BLEDevice.h>
2  #include <BLEServer.h>
3  #include <BLEUtils.h>
4  #include <BLE2902.h>
5  #include "DHT.h"
6
7  BLECharacteristic *characteristicTX; //através desse objeto iremos enviar
   dados para o client
8
9  bool deviceConnected = false; //controle de dispositivo conectado
10
11  const int LED = 2; //LED interno do ESP32
12
13  // See the following for generating UUIDs:
14  // https://www.uuidgenerator.net/
15
16  #define SERVICE_UUID "ab8028b1-198e-4351-b779-901fa0e0371e" // UART
   service UUID
17  #define CHARACTERISTIC_UUID_RX "4aca9682-9736-4e5d-932b-e9b31405049c"
18  #define CHARACTERISTIC_UUID_TX "0972128C-7613-4075-AD52-756F33D4DA91"
19  #define DHTPIN 4
20  //our sensor is DHT11 type
21  #define DHTTYPE DHT11
22  //create an instance of DHT sensor
23  DHT dht(DHTPIN, DHTTYPE);

```

Após isso, devem ser criados os callbacks, que são as funções que rodam quando a ESP recebe alguma resposta. A lógica pode ser de difícil compreensão, pois contém ponteiros, classes e objetos em C++, não se preocupe em entender o código por completo.

Primeiro vamos criar os ServerCallbacks, no nosso caso, executará uma função sempre que um dispositivo for conectado, ou desconectado da ESP.

```

1  //callback para receber os eventos de conexão de dispositivos
2  class ServerCallbacks: public BLEServerCallbacks {
3      void onConnect(BLEServer* pServer) {
4          deviceConnected = true;
5      };
6
7      void onDisconnect(BLEServer* pServer) {
8          deviceConnected = false;
9      }
10 };

```

Agora vamos criar os callbacks das características, sempre que é recebido um valor, ou seja, o cliente enviou algum comando, a função onWrite é executada (Write pelo ponto de visão do cliente). Nela é feita a leitura da mensagem, se a mensagem for "L1" o LED é ligado, se for "L0" o LED é desligado.

```

1  //callback para envio das características
2  class CharacteristicCallbacks: public BLECharacteristicCallbacks {

```

```

3      void onWrite(BLECharacteristic *characteristic) {
4          //retorna ponteiro para o registrador contendo o valor atual da
característica
5          std::string rxValue = characteristic->getValue();
6          //verifica se existe dados (tamanho maior que zero)
7          if (rxValue.length() > 0) {
8              Serial.println("*****");
9              Serial.print("Received Value: ");
10
11              for (int i = 0; i < rxValue.length(); i++) {
12                  Serial.print(rxValue[i]);
13              }
14
15              Serial.println();
16
17              // Realiza um comando de acordo com a mensagem recebida
18              if (rxValue.find("L1") != -1) {
19                  Serial.print("Turning LED ON!");
20                  digitalWrite(LED, HIGH);
21              }
22              else if (rxValue.find("L0") != -1) {
23                  Serial.print("Turning LED OFF!");
24                  digitalWrite(LED, LOW);
25              }
26
27              Serial.println();
28              Serial.println("*****");
29          }
30      }
31  };

```

Agora vamos fazer as configurações gerais, nosso **setup**, aqui vamos inicializar a porta serial, o sensor e o BLE.

```

1      void setup() {
2          Serial.begin(115200);
3
4          dht.begin();
5
6          pinMode(LED, OUTPUT);
7
8          // Create the BLE Device
9          BLEDevice::init("MINHA-ESP"); // nome do dispositivo bluetooth
10         // Cria o BLE Server
11         BLEServer *server = BLEDevice::createServer(); //cria um BLE server
12         server->setCallbacks(new ServerCallbacks()); //seta o callback do server
13         // Cria o BLE Service
14         BLEService *service = server->createService(SERVICE_UUID);
15         // Cria uma BLE Characteristic para envio de dados
16         characteristicTX = service->createCharacteristic(
17             CHARACTERISTIC_UUID_TX,
18             BLECharacteristic::PROPERTY_NOTIFY
19         );
20
21         characteristicTX->addDescriptor(new BLE2902());
22
23         // Cria uma BLE Characteristic para recebimento de dados
24         BLECharacteristic *characteristic = service->createCharacteristic(
25             CHARACTERISTIC_UUID_RX,

```



```

26                                     BLECharacteristic::PROPERTY_WRITE
27                                     );
28
29     characteristic->setCallbacks(new CharacteristicCallbacks());
30     // Start the service
31     service->start();
32     // Start advertising (ESP fica visível para outros dispositivos)
33     server->getAdvertising()->start();
34
35     Serial.println("Waiting a client connection to notify...");
36
37 }

```

O **loop** é muito simples, se existir um dispositivo conectado, a ESP fará a leitura do valor do sensor, converterá de **float** para **char** e envia isso para o cliente. Na porta serial são enviadas mensagens informando o valor que está sendo enviado.


```

1  void loop() {
2      //se existe algum dispositivo conectado
3      if (deviceConnected) {
4          float t = dht.readTemperature();
5
6          // Convertendo para char para poder enviar pelo BLE:
7          char txString[8]; // garantindo que é grande o suficiente
8          dtostrf(t, 1, 2, txString); // float_val, min_width,
           digits_after_decimal, char_buffer
9
10         characteristicTX->setValue(txString); //seta o valor que a
           característica notificará (enviar)
11         characteristicTX->notify(); // Envia o valor para o cliente
           (smartphone)
12
13         Serial.print("*** Sent Value: ");
14         Serial.print(txString);
15         Serial.println(" ***");
16     }
17     delay(1000);
18
19 }



```

5 Conectando com Smartphone


Para fazer a comunicação Bluetooth com a ESP, tanto leitura do sensor quanto envio de comandos, vamos utilizar um app no Smartphone:




Ao abrir o aplicativo, você terá na tela todos os dispositivos bluetooth encontrados pelo celular. A sua ESP deve aparecer, clique em CONNECT:




Ao conectar, a seguinte tela será mostrada, na qual aparecem os serviços desse dispositivo, CUSTOM SERVICE é o serviço que nós criamos, o UUID deve ser o mesmo que você configurou:




Ao clicar no serviço, as características serão apresentadas:



Ao clicar no círculo escrito **N** o valor que a ESP está enviando será mostrado na tela:



Ao clicar no círculo escrito **W** você poderá escrever um texto para enviar para a ESP. Lembrando do nosso código, ao enviar "L1" o LED deverá ser aceso:



Após seguir todos esses passos o seu sistema com comunicação Bluetooth Low Energy deve estar funcionando!